Define constraints/assumptions n parameters used. (Making them global by using $Assumptions = myAssumptions.)

In[184]:=

```
myAssumptions = {x + Re[k] > 0 && Re[1/θ] > -1, x ≥ 0, k ≥ 0, θ > 0, n ≥ k, p ≥ 0, p ≤ 1};

$Assumptions = myAssumptions
```

Out[185]=

$$\left\{x + \mathrm{Re}[k] > 0 \,\&\&\, \mathrm{Re}\left[\frac{1}{\theta}\right] > -1,\ x \geq 0,\ k \geq 0,\ \theta > 0,\ n \geq k,\ p \geq 0,\ p \leq 1\right\}$$

Illustrate shape k and scale θ formulation of Gamma as used by Mathematica

```
pdfGamma = Assuming[λ > 0, PDF[GammaDistribution[k, θ], λ] // Simplify]
Mean[GammaDistribution[k, θ]]
Variance[GammaDistribution[k, θ]]
```

Out[14]=

$$\frac{e^{-\frac{\lambda}{\theta}}\,\theta^{-k}\,\lambda^{-1+k}}{\mathrm{Gamma}[k]}$$

Out[15]=

$$k\,\theta$$

Out[16]=

$$k\,\theta^2$$

In[105]:=

```
$Assumptions = myAssumptions
FullSimplify[ Integrate[PDF[PoissonDistribution[λ], x]* pdfGamma, {λ, 0 , ∞}]]
```

Out[105]=

$$\left\{x + \mathrm{Re}[k] > 0 \,\&\&\, \mathrm{Re}\left[\frac{1}{\theta}\right] > -1,\ x \geq 0,\ k \geq 0,\ \theta > 0\right\}$$

Out[106]=

$$\frac{\theta^x\,(1+\theta)^{-k-x}\,\mathrm{Gamma}[k+x]}{x!\,\mathrm{Gamma}[k]}$$

Define my own version of the NBin() based on the $\lambda \sim \mathrm{Gamma}(k,\ \theta)$ and $x \sim \mathrm{Poisson}(\lambda)$ where x is motif_count

```
myNB[x_] = FullSimplify[ Integrate[PDF[PoissonDistribution[λ], x]* pdfGamma, {λ, 0 , ∞}]]
```

Out[44]=

$$\frac{\theta^x\,(1+\theta)^{-k-x}\,\mathrm{Gamma}[k+x]}{x!\,\mathrm{Gamma}[k]}$$

In[107]:=

```
mean = Simplify[Sum[myNB[x] * x, {x, 0, ∞}]]
var = FullSimplify[Sum[myNB[x] * (x - mean)^2, {x, 0, ∞}] // PowerExpand] // PowerExpand
```

Out[107]=

$$\frac{\theta\,\text{Gamma}[1+k]}{\text{Gamma}[k]}$$

Out[108]=

$k\,\theta\,(1+\theta)$

Compare our formulation of the NB with Mathematica's built in one.
We see that $n = k$ and $p = 1/(\theta+1)$

In[186]:=

```
PDF[NegativeBinomialDistribution[n, p], x] // Simplify
PDF[NegativeBinomialDistribution[k, 1/(θ + 1)], x] // Simplify


myNB[x]

(*Verify setting n = k and p = 1/(θ+1) and myNB[x] match*)
TrueQ[FullSimplify[% == %%]]
```

Out[186]=

$(1-p)^x\,p^n\,\text{Binomial}[-1+n+x, -1+n]$

Out[187]=

$\theta^x\,(1+\theta)^{-k-x}\,\text{Binomial}[-1+k+x, -1+k]$

Out[188]=

$$\frac{\theta^x\,(1+\theta)^{-k-x}\,\text{Gamma}[k+x]}{x!\,\text{Gamma}[k]}$$

Out[189]=

True

Verify formulation using Mathematica's NB distribution matches that of LindenAndMantyniemi2011

In[115]:=

```
sq = Superscript[σ, 2](*define σ^2 as a symbol rather than square of some σ*)
m = Mean[NegativeBinomialDistribution[n, p]]
v = Variance[NegativeBinomialDistribution[n, p]]
LMSol = (Solve[{m == μ, v == sq}, {n, p}] // FullSimplify)[[1]](* using σ^2*)
```

Out[115]=

σ^2

Out[116]=

$$\frac{n\,(1-p)}{p}$$

Out[117]=

$$\frac{n\,(1-p)}{p^2}$$

Out[118]=

$$\left\{n \to \frac{\mu^2}{-\mu+\sigma^2},\ p \to \frac{\mu}{\sigma^2}\right\}$$

In[194]:=

```
myNB[x]
```

Out[194]=

$$\frac{\theta^x\,(1+\theta)^{-k-x}\,\mathrm{Gamma}[k+x]}{x!\,\mathrm{Gamma}[k]}$$

Impose constraint where $\sigma^2 = a + b\,\mu$, the result is we can now formulate the NBin using μ and, in the future, μ_0 and t_0

In[195]:=

```
LMSolII = LMSol /. {sq → a + b μ} // FullSimplify

LMNB = (PDF[NegativeBinomialDistribution[n, p], x] /. LMSol /. {sq → a + b μ}) // FullSimplify //
   FunctionExpand

mySol = myNB[x] /. {k → n, θ → 1/p - 1} // FullSimplify

mySolII = mySol /. LMSolII // FullSimplify
```

Out[195]=

$$\left\{n \to \frac{\mu^2}{a + (-1 + b)\,\mu},\ p \to \frac{\mu}{a + b\,\mu}\right\}$$

Out[196]=

$$\frac{\left(\frac{\mu}{a+b\,\mu}\right)^{\frac{\mu^2}{a+(-1+b)\,\mu}} \left(1 - \frac{\mu}{a+b\,\mu}\right)^{x} \mathrm{Gamma}\left[x + \frac{\mu^2}{a+(-1+b)\,\mu}\right]}{\mathrm{Gamma}[1 + x]\ \mathrm{Gamma}\left[\frac{\mu^2}{a+(-1+b)\,\mu}\right]}$$

Out[197]=

$$\frac{(1 - p)^{x}\ p^{n}\ \mathrm{Gamma}[n + x]}{x!\ \mathrm{Gamma}[n]}$$

Out[198]=

$$\frac{\left(\frac{\mu}{a+b\,\mu}\right)^{\frac{\mu^2}{a+(-1+b)\,\mu}} \left(1 - \frac{\mu}{a+b\,\mu}\right)^{x} \mathrm{Gamma}\left[x + \frac{\mu^2}{a+(-1+b)\,\mu}\right]}{x!\ \mathrm{Gamma}\left[\frac{\mu^2}{a+(-1+b)\,\mu}\right]}$$

In[199]:=

```
(*Verify setting n = k and p = 1/(θ+1) and myNB[x] match*)
TrueQ[FullSimplify[mySolII == LMNB, Assumptions → {x > 0, b > 1}]]
```

Out[199]=

```
True
```

Conclusion

In Mathematica, we can formulate our PDF for a data point $\Big|\ \mu = \mu(t) = \begin{cases} \mu_0 & \text{if } t < t_0 \\ \mu_0 - b\,(t - t_0) & \text{else} \end{cases}$; where $b = \frac{\mu_0}{t_{max} - t_0}$

In[201]:=

```
Simplify[PDF[NegativeBinomialDistribution[μ[t]^2/(a + (b - 1) μ[t]), μ[t]/(a + b μ[t])], x]]
```

Out[201]=

$$\text{Binomial}\left[-1 + x + \frac{\mu[t]^2}{a + (-1 + b)\,\mu[t]}, -1 + \frac{\mu[t]^2}{a + (-1 + b)\,\mu[t]}\right] \left(\frac{\mu[t]}{a + b\,\mu[t]}\right)^{\frac{\mu[t]^2}{a+(-1+b)\,\mu[t]}} \left(1 - \frac{\mu[t]}{a + b\,\mu[t]}\right)^x$$